

Lesson 10

Differentials, Linear and Quadratic Approximations

Initializations

```
> restart;  
with(liesymm):  
>
```

10.1 Differentials

If $y = f(x)$, then the differential dy is defined by

$$dy = \frac{dy}{dx} dx = f'(x) dx$$

Maple's routine for computing differentials is accessed by the d command in the **liesymm** package.

Examples

Example 10.1.1

Let $y = x^3 \sin 2x$, compute dy .

Solution

First we code the expression $y = x^3 \sin 2x$, and set up the independent variable x for use by the d routine.

```
> y:=x^3*sin(2*x);  
y := x3 sin(2 x) (2.1.1.1)
```

```
> setup([x]);  
[x] (2.1.1.2)
```

Compute the differential dy by evaluating $d(y)$.

```
> dy:=d(y);  
dy := 3 x2 d(x) sin(2 x) + 2 x3 cos(2 x) d(x) (2.1.1.3)
```

If so desired, this result can be rewritten in the standard form

$$dy = \frac{dy}{dx} dx$$

by collecting all terms that contain a dx .

```
> collect(dy, d(x));  
(3 x2 sin(2 x) + 2 x3 cos(2 x)) d(x) (2.1.1.4)
```

```
>
```

Example 10.1.2

An aluminum ball with diameter 2 inch has a surface which is 0.1 inch thick. Use differentials to estimate the amount of aluminum used to construct the ball.

Solution

Code the volume V of a sphere as a function of the radius r , and compute dV .

```
> V:=4/3*Pi*r^3;
```

$$V := \frac{4}{3} \pi r^3 \quad (2.1.2.1)$$

```
> setup([r]);
```

$$[r] \quad (2.1.2.2)$$

```
> dV:=d(V);
```

$$dV := 4 \pi r^2 d(r) \quad (2.1.2.3)$$

The desired estimate for the amount of aluminum used, is obtained by substituting $r = 2$ and $dr = 0.1$ in the formula for dV .

```
> estimate:=evalf(subs({r=2, d(r)=0.1}, dV));
```

$$\text{estimate} := 1.256637062 \quad (2.1.2.4)$$

```
>
```

10.2 Linear and Quadratic Approximations

Examples

Example 10.2.1

Compute the linear and quadratic approximations of $f(x) = \sqrt{x+1}$ at $x = 3$. Determine for which values of x each of these approximations is accurate to within 0.1.

Solution

As derived in-class, the linear (LA) and quadratic (QA) approximations of $f(x)$ at $x = a$, are given by

$$LA = f(a) + f'(a)(x-a)$$

and

$$QA = f(a) + f'(a)(x-a) + \frac{1}{2}f''(a)(x-a)^2$$

Maple makes quick work of these computations.

```
> f:=x->sqrt(x+1);
```

$$f := x \rightarrow \sqrt{x+1} \quad (3.1.1.1)$$

```
> LA:=f(3)+D(f)(3)*(x-3);
```

$$LA := \frac{5}{4} + \frac{1}{4}x \quad (3.1.1.2)$$


```
> QA:=f(3)+D(f)(3)*(x-3)+(D@@2)(f)(3)/2*(x-3)^2;
```

$$(3.1.1.3)$$

$$QA := \frac{5}{4} + \frac{1}{4}x - \frac{1}{64}(x-3)^2 \quad (3.1.1.3)$$

Next we address the accuracy issue. First make a sketch.

```
> plot([f(x)+0.1, f(x)-0.1, LA, QA], x=-0.3..8, color=[gold, gold, blue, red], thickness=2);
```



We read from the picture that the linear approximation is within the prescribed bound when x is approximately on the interval

$$[0.89, 5.9]$$

A similar result can be obtained for the quadratic approximation.

$$[0.07, 7.33]$$

More precise endpoints for these intervals can be obtained by solving the inequalities

$$|f(x) - LA| < 0.1$$

and

$$|f(x) - QA| < 0.1$$

```
> solve(abs(f(x)-LA)<=0.1, x);
      RealRange(0.8701778719, 5.929822128)      (3.1.1.4)
```

```
> solve(abs(f(x)-QA)<=0.1, x);
      RealRange(3., 7.387051747), RealRange(0.06265310308, Open(3.))      (3.1.1.5)
```

```
>
```

Clearly, the quadratic approximation is accurate on a larger interval than the linear approximation.